
NURBS, or Non Uniform Rational B Splines, are a curved surface representation commonly used in computer aided design and digital content creation. They define a mathematical surface as a collection of points computed from a tensor product of the NURBS basis functions, control vertices, and weights:

Where Bi,m(u), Bj,n are the NURBS basis functions in u and v, Pi,j are the control vertices, and wi,j are the weights. The NURBS basis functions, Bi,k are defined recursively in terms of a knot vector ti:

This recursive representation gives a great deal of flexibility, allowing arbitrary surface order and knot vectors, enabling a single NURBS surface to contain many contiguous patches. However, this recursive representation is also expensive to compute, so a NURBS surface is often converted into multiple Bezier patches before being tessellated.

In this implementation, we present an efficient method for directly tessellating NURBS surfaces using the NVIDIA CUDA computing API.

For our implementation of CUDA tessellation of NURBS, there are 5 main steps:

1) Compute Edge Tessellation Factors

2) Pre-Compute Polynomial Basis Coefficients

3) Compute U,V Tessellation Pattern

4) Compute Positions, Normals
5) Triangulate Surface
We provide input in the form of an array of NURBS patches, with each patch defined as follows:
struct NURBSPatch

{

  float4 cp[32*32];       //Control Vertices

  float ku[64];           //Knot Vector in U

  float kv[64];           //Knot Vector in V

  float tessFactor[4];    //Tessellation factor for each edge

  int nCPsU;              //Number of control vertices in U

  int nCPsV;              //Number of control vertices in V

  int degreeU;            //Degree of curve in U

  int degreeV;            //Degree of curve in V

  int nKnotsU;            //Number of knots in U

  int nKnotsV;            //Number of knots in V

  NURBS_BOUNDARY_CONDITION boundaryConditionU;
  NURBS_BOUNDARY_CONDITION boundaryConditionV;

  float4* vertex;         //Ptr into the dest VBO for writing vert positions
  float3* normal;         //Ptr into the dest VBO for writing vert normals
  unsigned int *index;    //Ptr into the dest VBO for writing tri indices

  unsigned int firstVertexIndex;  //First vertex index used by this patch

};

The output will be VBOs for positions, normal, and triangle indices, with each patch being assigned a segment of the VBOs into which to tessellate.
1) Compute Edge Tessellation Factors
First we invoke a CUDA kernel, ComputeNURBSTessellationLevel_Kernel(…), to compute the tessellation factors for each edge of each surface, in a manner that ensures adjoining patches will have the same tessellation factor along the shared edge. For this implementation, a simple computation is performed, summing the distances between the control vertices, multiplying by a constant, and dividing by the camera distance. This gives a relatively constant tessellated triangle size onscreen:
tessLevel = C*(Σi length(CP[i+1] – CP[i])) / distanceToCamera
Once the tessellation level is computed for each edge, we compute the number of vertices and indices in the patch, atomically add them to global counters, and use the resulting index to compute where in the destination VBOs the tessellated positions, normal, and indices will be written:

patch->tessFactor[0] = ComputeTessFactorCU(…);

patch->tessFactor[1] = ComputeTessFactorCU(…);

patch->tessFactor[2] = ComputeTessFactorCU(…);

patch->tessFactor[3] = ComputeTessFactorCU(…);

int nVertices = numberOfVerticesCU(*patch);

int nIndices = numberOfIndicesCU(*patch);

int vIndex = atomicAdd(&g_VertexCount, nVertices);

int iIndex = atomicAdd(&g_IndexCount, nIndices);

patch->vertex = (float4*)(g_VertexStart + sizeof(float4)*vIndex);

patch->normal = (float3*)(g_NormalStart + sizeof(float3)*vIndex);

patch->index = (unsigned int*)(g_IndexStart + sizeof(unsigned int)*iIndex);

patch->firstVertexIndex = vIndex;
Now we pass the array of patches to the second CUDA kernel, tesselateNURBSPatches_Kernel(…) which will perform the tessellation and write the output vertex positions, normals, and triangle indices to the VBO locations assigned above.
2) Pre-Compute Polynomial Basis Coefficients

Because NURBS basis functions are normally defined recursively, they can be very expensive to evaluate per tessellated vertex. In order to perform a more efficient evaluation, we define the NURBS basis functions as a polynomial series and pre-compute the polynomial coefficients. 

The computation of the polynomial coefficients is defined recursively, as below:

If you look closely at the above, you will notice that the coefficients are dependent only on the knot vector, and not on t itself (Ci,0,0 = Bi,0 := (0 or 1)), and thus only need to be computed on the knot spans, greatly reducing the amount of computation required.
Using a polynomial series also makes computation of the derivatives (and thus normals) much easier:

To perform the computation of the coefficients efficiently in parallel in our CUDA kernel, we use one thread to compute each coefficient and store it in a shared memory array:
__shared__ float CU[6][6][32]; //Good up to 6th order, 32 knot spans max
First we figure out the number of coefficients we need to compute:

unsigned int nUCoefficients = nSpansU*orderU*orderU;

We then loop through them, incrementing inc by blockDim.x:

for(int inc = 0; inc < nUCoefficients; inc += blockDim.x)

We then add threadIdx.x to inc to get a unique linear index.

int idx = inc + threadIdx.x;

From this, we compute indices to cover the range of coefficients we need to generate: 

int uSpan = idx%(nSpansU);

int cpIdxU = (idx/nSpansU)%orderU;

int k = (idx/nSpansU)/orderU;

int i = cpIdxU + uSpan - degreeU;

And use these as indices for the shared memory location to store the coefficient, and as inputs to the recursive function to compute the coefficient:

CU[k][cpIdxU][uSpan] = ComputeCoefficient(uKnotVector, uSpan, i, degreeU, k);
We then do the same for the basis coefficients in V, storing them in shared memory as well
3) Compute U,V Tessellation Pattern

The next step is to compute the tessellation pattern in u and v. It is desirable to have symmetric, fractional tessellation that can vary across the edges of the patch in order to have seamless, continuously varying tessellation across the whole model.

[image: image7.jpg]



To accomplish the computation of the u,v tessellation pattern in parallel, we first compute the patch tessellation factors and the resultant number of tessellated vertices that will be created:

//Compute tessellation level for interior of patch - Max of opposing edges 

float maxTessU = max(patch->tessFactor[1], patch->tessFactor[3]);

float maxTessV = max(patch->tessFactor[0], patch->tessFactor[2]);

float ceilTessU = ceil(maxTessU);

ceilTessU += ((int)ceilTessU)&1;
//round up to even number of U intervals
float ceilTessV = ceil(maxTessV);

ceilTessV += ((int)ceilTessV)&1;
//round up to even number of V intervals
int nVerticesU = ceilTessU + 1;

int nVerticesV = ceilTessV + 1;

int nPatchVertices = nVerticesU * nVerticesV;
We then loop over all the patch vertices, incrementing by blockDim.x:

for(int inc = 0; inc < nPatchVertices; inc += blockDim.x)
And compute a unique linear index for each vertex:
int idx = inc + threadIdx.x;

We then decompose this index into a 2-dimensional integer index
int idxU = idx % nVerticesU;

int idxV = idx / nVerticesU;
And then use these integer indices to compute floating point u and v coordinates on the tessellated surface, using the below function:

__device__ float tessCoord(float tessFactor, int idx, float u0, float un)

{

  float u;

  float ceilTess = ceil(tessFactor);

  ceilTess += ((int)ceilTess)&1;

  float w = (un - u0)/tessFactor;

  if(idx < ceilTess*0.5f - EPSILON) 


  u = u0 + (float)idx*w;

  else if(idx > ceilTess*0.5f + EPSILON)


  u = un - EPSILON - (ceilTess - (float)idx)*w;

  else


  u = u0 + 0.5f*(un - u0);

  return u;

}
Essentially, this computes the coordinates in from the edges, in increments of w, leaving a fractional interval between the innermost coordinate and the center, as shown in the earlier diagram.
Notice in this diagram that the top and right edges are tessellated differently than the rest of the surface. This is accomplished by aliasing several points to the same u,v by computing an adjusted integer index for the edges as follows:

if(idxV == 0) //Bottom Edge

{
  float ceilTessU1 = ceil(patch->tessFactor[1]);

  ceilTessU1 += ((int)ceilTessU1)&1;
  int idxU2;

  if(idxU < ceilTessU*0.5f){
    //Scale index by edge/interior tess, round down

    idxU2 = (int)floor((ceilTessU1/ceilTessU)*(float)idxU);     

  }
  else{
    //Scale index by edge/interior tess, round up
    idxU2 = (int)ceil((ceilTessU1/ceilTessU)*(float)idxU);

  }
  u = tessCoord(patch->tessFactor[1], idxU2, u0, un);

}

This process is replicated for each of the 4 edges, even though 2 of the edges will tessellate to the same value as the interior of the patch. Note that for this scheme to work, the interior tessellation must always be greater than or equal to the edge tessellation, which is why we compute the patch tessellation as the max of the opposing edges
4) Compute Position, Normal for Each Tessellated Vertex
We then compute the position and normal for each vertex. We start by determining which u and v span that the (u,v) point lies on:

unsigned int uSpan = findSpan(u, patch->ku, patch->nKnotsU);    

unsigned int vSpan = findSpan(v, patch->kv, patch->nKnotsV);
We then initialize the vertex position, tangent, and bi-tangent to zero and loop over all the relevant control vertices:

float4 position = ZeroFloat4();

float4 tangent = ZeroFloat4();

float4 biTangent = ZeroFloat4();

float sumN = 0.0f;

//loop over the relevant control points

for(int j = vSpan - degreeV; j < vSpan + 1; j++)

{

  for(int i = uSpan - degreeU; i < uSpan + 1; i++)

  {
We then pick the control point to use, modulating by the number of control points in u and v in order to handle closed surfaces where the span wraps around:

float4 cp = patch->cp[(i%patch->nCPsU) + (j%patch->nCPsV)*patch->nCPsU];
And then compute the value for the NURBS basis function, using Horner’s rule to compute the polynomial series from our pre-computed polynomial coefficients stored in shared memory:

int cpIdx = i - (uSpan - degreeU);

float Bu = CU[degreeU][cpIdx][uSpan];

float dBu_du = Bu*degreeU;

for(int k = degreeU-1; k > 0; k--)

{

  float C = CU[k][cpIdx][uSpan];

  Bu = u * Bu + C;

  dBu_du =  u*dBu_du + k * C;

}

Bu = u * Bu + CU[0][cpIdx][uSpan];
Note that we also efficiently compute the NURBS basis for the derivative (dBu_du) at the same time

We then repeat for the V basis, then use both these values to compute this control vertex’s contribution to the position, tangent, and bi-tangent:

position += (Bu*Bv*cp.w) * cp;

tangent += (dBu_du*Bv*cp.w) * cp;

biTangent += (Bu*dBv_dv*cp.w) * cp;

sumN += Bu*Bv*cp.w;
Once we have computed the contribution from all the control points, we normalize the vertex position:

float ooSumN = (1.0f/sumN);

patch->vertex[idx] = position * ooSumN;

and compute the vertex normal:
tangent = tangent * ooSumN;

biTangent = biTangent * ooSumN;

patch->normal[idx] =

normalize(cross(normalize(tangent),normalize(biTangent)));
5) Triangulate Patch
The final step is to triangulate the patch, compute the triangle indices and write them into the index VBO.

First we compute how many indices will be created for this patch:

int nQuadsU = ceilTessU;

int nQuadsV = ceilTessV;

int nPatchQuads = nQuadsU * nQuadsV;
Then we perform the triangulation in parallel by looping over all the indices, incrementing by blockDim.x:

for(int i = 0; i < nPatchQuads; i += blockDim.x)
We then compute a unique linear index for each quad to be triangulated:

int idx = i + threadIdx.x;
and use this to compute 2-D indices for each quad

int idxU = idx % nQuadsU;

int idxV = idx / nQuadsU;

We then compute an offset for which vertex in the final VBO to start with for this patch

int offset = patch->firstVertexIndex;
And, finally, compute the triangulated indices:

index[idx*6] = offset + idxU + idxV*nVerticesU;

index[idx*6+1] = offset + idxU + (idxV+1)*nVerticesU
index[idx*6+2] = offset + idxU+1 + (idxV+1)*nVerticesU;
index[idx*6+3] = offset + idxU + idxV*nVerticesU;
index[idx*6+4] = offset + idxU+1 + (idxV+1)*nVerticesU;
index[idx*6+5] = offset + idxU+1 + idxV*nVerticesU;

